
dado. Pero el programa puede necesitar desplegar copias múltiples de un caracter todo de 
negro para números mayores a 10; para esto se usa el caracter personalizado 3.   
 

 

Figura 7-2 
Libreria de Definiciones 
para el caracter 
personalizado 2 

 

Programa Ejemplo: HorizBarGraph.bs2 

√ Introduzca, salve y corra HorizBarGraph.bs2. 
√ Haga Click en la Ventana de Transmisión de la Terminal de Depuración. 
√ Escriba un 8, luego presione Enter. 
√ Revise la gráfica de barra y asegúrese que el caracter mas a la izquierda esta 

negro y que el siguiente muestra Char3. 
√ Intente los valores mostrados en la Figura 7-1. 
√ Intente 45, 46, 47, 48, 49, 50, 51, 52. 
√ Experimente con los límites como 0, 80. 
√ Intente valores fuera de los límites. ¿Qué pasa? 

 
' -----[ Titulo ]------------------------------------------------------------- 
' Sensores Inteligentes y sus Aplicaciones - HorizBarGraph.bs2 
' Despliega valores introducidos en la Ventana de Recepcion de la Terminal de 
' Depuracion como datos de una grafica de barra horizontal en la LCD. 
 
' {$STAMP BS2}                      ' Dispositivo Objeto  = BASIC Stamp 2 
' {$PBASIC 2.5}                     ' Lenguaje            = PBASIC 2.5 
 
' -----[ Pines E/S]----------------------------------------------------------- 
 
LcdPin         PIN     14                   ' pin E/S conectado a RX de la LCD 

' -----[ Constantes ]--------------------------------------------------------- 
 
T9600        CON     84                    ' True, 8-bits, no parity, 9600 
 
LcdCls         CON     12                    ' Form feed -> limpia pantalla 
LcdCr          CON     13                    ' Retorno de carro 
LcdOff         CON     21                    ' Apaga la pantalla 
LcdOn          CON     22                    ' Enciende la pantalla 
Line0          CON     128                   ' Linea 0, caracter 0 
Line1          CON     148                   ' Linea 1, caracter 0 
Define         CON     248                   ' direccion define caract pers 0 
 
' -----[ Variables ]---------------------------------------------------------- 
 
custChar       VAR     Nib                 ' selector de carcter personalizado  
index          VAR     Nib                   ' variable indice Eeprom  
dotLine        VAR     Byte                  ' linea punteada de 5-pixel 
cursor         VAR     Nib                   ' Colocacion de Cursor  
value          VAR     Byte                  ' Valor a ser graficado 
charCnt        VAR     Byte                  ' variable para conteo Caracter  
line           VAR     Byte                  ' Linea0 o Linea1 
 
' -----[ inicialización ]----------------------------------------------------- 
 
PAUSE 100                                    ' Estabiliza fuente de poder 
SEROUT LcdPin, T9600, [LcdOn, LcdCls]        ' Initializa LCD 
PAUSE 5                                   ' 5 ms retraso para limpiar pantalla 
 
custChar = 3                             ' Selecciona caracter personalizado 3 
dotLine = %11111                    ' Oscurece todos los pixeles en cada linea 
GOSUB Def_Horiz_Bar_Char                     ' Subrutina define Caracter  
 
line = Line0                                 ' BarGraph en Linea 0. 
 
DEBUG "Enter valores (0 to 80)", CR 
 
' -----[ Main Rutina ]------------------------------------------------------- 
 
DO                                           ' Ciclo Principal 
 
  DEBUG ">" 
  DEBUGIN DEC value                          ' Valor en ventana de Transmision  
  GOSUB Bar_Graph                            ' Despliega como grafica de barra  
 
LOOP                                         ' Repite ciclo principal 
 
' -----[ Subrutina - Bar_Graph ]---------------------------------------------- 
 
Bar_Graph: 
 
  ' Llena desde la izquierda con barras negras 

  value = value MAX 80                       ' valor Limite - 0 a 80 
  charCnt = value / 5                        ' número de barras negras  
  custChar = 3                               ' Escoge caracter pers. negro 
 
  IF charCnt > 0 THEN                        ' Si escribe  barras negras… 
    FOR cursor = 0 TO charCnt - 1          ' Escribe charCnt - 1 barras negras  
      GOSUB Disp_Cust_Char                   ' Escribe la barra negra  
    NEXT 
  ENDIF 
 
  ' Despliega caracter pers. 2 con cierto número de columnas negras. 
 
  cursor = charCnt                           ' Coloca cursor 
  custChar = value // 5                      ' ¿Cuantos quintos de barra? 
  ' Escoge patron de bits para la definicion de caracter personalizado  
  LOOKUP custChar, 
         [%00000, %10000, %11000, %11100, %11110], 
         dotLine 
  custChar = 2                         ' Establece caracter personalizado en 2 
  GOSUB Def_Horiz_Bar_Char                  ' Define el caracter personalizado 
  GOSUB Disp_Cust_Char                   ' Despliega el caracter personalizado 
 
  ' Sobreescribe todo a la derecha con espacios. 
 
  IF (charCnt + 1) < 15 THEN      ' ¿Caracteres Parciales a la izq de char 15? 
    FOR cursor = (charCnt + 1) TO 15        ' Llena a la derecha con " " 
      SEROUT LcdPin, T9600, 
             [line + cursor, " "] 
    NEXT 
  ELSEIF value = 80 THEN                    ' Caso especial: value = 80 
    SEROUT LcdPin, T9600, 
           [line + cursor, 3] 
  ELSEIF charCnt = 14 THEN                  ' Caso especial: 75 <= value <= 80 
    SEROUT LcdPin, T9600, [line + 15, " "] 
  ENDIF 
 
  RETURN 
 
' -----[ Subrutina - Def_Horiz_Bar_Char ]------------------------------------ 
 
Def_Horiz_Bar_Char: 
 
  SEROUT LcdPin, T9600,                   ' Define caracter personalizado 
         [Define + custChar] 
  FOR index = 0 TO 7                      ' 7 bytes, define 7 lineas punteadas  
    SEROUT LcdPin, T9600, [dotLine]       ' Manda a la LCD 
  NEXT 
 
  RETURN 

' -----[ Subrutinas - Disp_Cust_Char ]--------------------------------------- 
 
' Esta subrutina despliega un caracter personalizado. La variable line puede 
' establecerse en Line0 o Line1, y la variable cursor puede establecerse a un 
' valor entre 0 y 15. La variable custChar selecciona uno de los caracteres 
' personalizados de la LCD y debe ser establecida a un valor entre 0 y 7. 
 
Disp_Cust_Char: 
 
  SEROUT LcdPin, T9600,                    ' Escribe caracter personalizado 
         [line + cursor, custChar] 
  RETURN 
 

Como trabaja HorizBarGraph.bs2  

El caracter personalizado 2 de la LCD es reservado para desplegar una de las 5 diferentes 
definiciones de caracteres personalizados, pero el caracter personalizado 3 siempre usará 
la misma definición, tener todos sus pixeles negros. Con esto en mente, un bloque de 
código fue agregado a la sección de inicialización que define el caracter personalizado 3 
de la LCD usando la subrutina Def_Horiz_Bar_Char.  Esta subrutina hace 8 copias 
idénticas del renglón de 5 pixeles guardados por la variable dotLine para construir el 
caracter personalizado. Después de esto, la variable line se establece para colocar el 
cursor en el renglón de arriba y el programa despliega un aviso al usuario "Enter values 
(0 to 80)", seguido de un retorno de carro. 

 
custChar = 3                      ' Selecciona caracter personalizado 3 
dotLine = %11111             ' Oscurece todos los pixeles en cada linea 
GOSUB Def_Horiz_Bar_Char             ' Subrutina define Caracter 
 
line = Line0                         ' BarGraph en Linea 0. 
 
DEBUG "Enter valores (0 to 80)", CR 
 

La Rutina Principal es un DO...LOOP que repetídamente despliega el aviso ">", y luego 
obtiene valores decimales introducids en la Ventana de Transmisión de la Terminal de 
Depuración. Luego llama a la  subroutine Bar_Graph.  

 
' -----[ Rutina Principal ]-------------------------------------------- 
 
DO                                 ' Ciclo Principal 
 
  DEBUG ">" 
  DEBUGIN DEC value                ' Valor en ventana de Transmision 
  GOSUB Bar_Graph                  ' Despliega como grafica de barra 
 
LOOP                               ' Repite ciclo principal 

La subrutina Bar_Graph toma lo que esté guardado en la  variable value y lo representa 
en la LCD con una pantalla de gráfico de barra. Esta subrutina confía en las subrutinas 
Def_Cust_Char y Disp_Cust_Char de la actividad previa. La subrutina Bar_Graph 
consiste en 3 pasos mayores:   
 

1. Llena cualesquiera caracteres negros de izquierda a derecha. Por ejemplo, si la 
variable value es establecida en 28, 5 caracteres LCD (con 5 lineas negras 
verticales cada uno) deben ser oscurecidos. 

2. En el ejemplo, el sexto caracter LCD tendrá 3 líneas verticales. Recuerde, el 
caracter personalizado 2 es usado para desplegar 1 de las 5 definiciones de 
caracter (Figura 7-2). El número de columnas negras en el caracter es el residuo 
de value // 5. Este resultado selecciona un patrón de bits a partir de una tabla 
LOOKUP y lo copia a la dotLine. Luego, custChar es igualada a 2 y la subrutina 
Def_Horiz_Bar_Char copia este patrón de bits a los 8 renglones en el caracter. 
Ya que el caracter fue redefinido puede ser entonces escrito.   

3. Todos los caracteres no necesarios para representar el valor en la gráfica de barra 
tienen que ser limpiados con el  caracter espacio " ". En este ejemplo value = 
28, esto es, limpiar todo entre la derecha del sexto caracter y el 15avo caracter.  
Si bien esto es sola y absolutamente necesario si value anterior es menor que 
value actual, el programa lo hace todo el tiempo a través del ciclo. 

 
El primer paso en la subrutina Bar_Graph es llenar caracteres negros. value primero es 
forzada a 80 o menos.  Luego, charCnt guarda el número de caracteres negros que tienen 
que ser escritos, que es 1/5 de value. custChar tiene que ser igualada a 3, puesto que el 
caracter personalizado 3 guarda el carácter negro.  Si charCnt es mayor que 0, significa 
que habrán algunos  caracteres completamente negros que tienen que ser escritos, y un 
FOR...NEXT repetídamente llama a la subrutina Disp_Cust_Char. Recuerde, esta 
subrutina depende de 2 variables: cursor y custChar. El valor de charCnt fué 
establecido antes del ciclo FOR...NEXT, y la variable cursor es la variable index del ciclo 
FOR...NEXT.  Cada vez que se cumple el ciclo FOR...NEXT, cursor se incrmenta en 1, lo 
que  causa que la subrutina Disp_Cust_Char coloque el cursor un paso a la derecha cada 
vez que es llamada, y por lo tanto llenando caracteres negros de izquierda a derecha. 

 
' -----[ Subrutina - Bar_Graph ]----------------------------------- 
 
Bar_Graph: 
 
  ' Llena desde la izquierda con barras negras 

  value = value MAX 80             ' valor Limite - 0 to 80 
  charCnt = value / 5              ' número de barras negras  
  custChar = 3                     ' Escoge caracter pers. negro 
 
  IF charCnt > 0 THEN              ' Si escribe  barras negras… 
    FOR cursor = 0 TO charCnt - 1  ' Escribe charCnt - 1 barras negras 
      GOSUB Disp_Cust_Char         ' Escribe la barra negra 
    NEXT 
  ENDIF 
 

El segundo paso es desplegar el caracter que está parcialmente negro. cursor = 
charCnt asegura que el cursor ahora está justo a la derecha de los caracteres negros que 
fueron escritos con un FOR...NEXT en el paso anterior. Luego, custChar = value // 5 
establece custChar en el residuo de value ÷ 5.  Por ejemplo, si value es 28, el residuo de 
28 ÷ 5 es 3.  Si custChar = 3, una tabla lookup guarda %11100 en dotLine. La 
subrutina Def_Horiz_Bar_Char necesita saber 2 cosas:  dotLine y custChar. Estamos 
usando y re-usando el caracter personalizado 2 para definir y redefinir el caracter 
parcialmente oscurecido. Entonces, antes de llamar a Def_Horiz_Bar_Char, custChar 
necesita ser cambiada de 3 a 2 con custChar = 2.  Entonces, Def_Horiz_Bar_Char 
puede ser llamada para definir el caracter personalizado, seguida por Disp_Cust_Char 
para desplegarla. 

 
  ' Despliega caracter pers. 2 con cierto número de columnas negras. 
 
  cursor = charCnt                 ' Coloca cursor 
  custChar = value // 5            ' ¿Cuantos quintos de barra? 
  ' Escoge patron de bits para la definicion de caracter personalizado 
  LOOKUP custChar, 
         [%00000, %10000, %11000, %11100, %11110], 
         dotLine 
  custChar = 2                  ' Establece caracter personalizado en 2 
  GOSUB Def_Horiz_Bar_Char           ' Define el caracter personalizado 
  GOSUB Disp_Cust_Char            ' Despliega el caracter personalizado 
 

La subrutina Def_Horiz_Bar_Char que es llamada después de que se establece el valor 
de las variables dotLine y custChar hace innecesario que se guarden caracteres 
personalizados en la EEPROM. Por esta razón, si quiere crear un caracter personalizado 
con varias columnas de pixeles negros, todo lo que tiene que hacer  es enviar a la LCD el 
mismo valor binario, 8 veces en un reglón.  La variable dotLine es la que guarda la 
definición binaria para los renglones en el caracter personalizado parcialmente llenado. Si 
dotLine es %11000, las 2 columnas izquierdas de pixeles se harán negras. Si dotLine 
es %11100, las 3 columnas izquierdas de pixeles se harán negras, etc. 

Def_Horiz_Bar_Char: 
 
  SEROUT LcdPin, T9600,            ' Define caracter personalizado 
         [Define + custChar] 
  FOR index = 0 TO 7               ' 7 bytes, define 7 lineas punteadas 
    SEROUT LcdPin, T9600, [dotLine] ' Manda a la LCD 
  NEXT 
 
  RETURN 
 

Sobreescribir cualesquiera caracteres negros a la derecha del caracter desplegado en el 
paso 2 no involucra más caracteres personalizados puesto que el caracter espacio " " hace 
un buen trabajo para borrar las cosas. En la mayoría de los casos, un ciclo FOR...NEXT que 
escriba de (charCnt + 1) TO 15 limpia todo a la derecha. Sin embargo, hay algunas 
circunstancias especiales que ocurren cuando el 15avo caracter tiene una o más columnas 
negras. Si value = 80, una parte anterior del programa imprimirá un carácter negro en 
la posición-0.  Si la posición-0 fuera negra, el bloque de código ELSEIF value = 80 
reemplaza ese caracter vacío con uno negro. También, si charCnt es 14, un solo caracter 
vacío tiene que ser escrito en la posición 15.   

 
  ' Sobreescribe todo a la derecha con espacios. 
 
  IF charCnt + 1 < 15 THEN ' ¿Caracteres Parciales a la izq de char 15? 
    FOR cursor = (charCnt + 1) TO 15  ' Llena a la derecha con " " 
      SEROUT LcdPin, T9600, 
             [Line0 + cursor, " "] 
    NEXT 
  ELSEIF value = 80 THEN             ' Caso especial: valor = 80 
    SEROUT LcdPin, T9600, 
           [Line0 + cursor, 3] 
  ELSEIF charCnt = 14 THEN           ' Caso especial: 75 <= valor <= 80 
    SEROUT LcdPin, T9600, [Line0 + 15, " "] 
  ENDIF 
 
  RETURN 

Su Turno - Desplegado Gráfico de la Distancia al Sensor Ping))) 

Es sencillo desplegar una gráfica de barra horizontal que indique hasta 80 cm con el 
sensor Ping))) y la LCD Serial Parallax. El truco es tomar componentes de 
PingMeasureCm.bs2 e incorporarlos en una copia de HorizBarGraph.bs2 de esta 
actividad. 
 

√ Siga las instrucciones para conectar el sensor Ping))) y la LCD Serial a su tarjeta 
(Capítulo 2, actividad #4, página 51). 

√ Abra PingMeasureCm.bs2 (Capítulo 2, actividad #2 página 48) en el Editor de 
BASIC Stamp. 

√ Abra HorizBarGraph.bs2 (esta actividad), y sálvelo como PingBarGraph.bs2. 
√ Sombree y copie las directivas CON y VAR de PingDistance.bs2, y péguelas en las 

secciones CON y VAR en PingBarGraph.bs2. 
√ Reemplace los comandos DEBUG y DEBUGIN en el DO...LOOP de la Rutina 

principal de PingBarGraph.bs2 con los comandos en el DO...LOOP de 
PingMeasureCm.bs2.   

√ Agrege un comando justo antes de GOSUB Bar_Graph que iguale la variable 
value a cmDistance.   

√ Mueva el comando PAUSE 100 para que quede justo antes del ciclo.   
 
Ahora, la rutina principal en PingBarGraph.bs2 debe verse así: 

 
' -----[ Rutina Principal ]-------------------------------------------- 
 
DO                                    ' Ciclo Principal 
 
  PULSOUT 15, 5 
  PULSIN 15, 1, time 
 
  cmDistance = CmConstant ** time 
 
  DEBUG HOME, DEC3 cmDistance, " cm" 
 
  value = cmDistance 
 
  GOSUB Bar_Graph                     ' Despliega como grafica de barra 
 
  PAUSE 100 
 
LOOP                                  ' Repite ciclo principal 
 

√ Salve el programa modificado, pruébelo, y verifique que trabaja. 
 
También puede reemplazar el comando DEBUG con un comando SEROUT que despliegue 
las mediciones en la línea inferior.  Recuerde que tendrá que enviar a la LCD un código 
de control para colocar el cursor en la Línea 1 caracter-0, en vez de usar HOME. 
 

√ ¡Inténtelo! 



Capítulo 7: Gráficos de Barra en LCD para Distancias e inclinaciones · Page 281 

ACTIVIDAD #3: GRAFICA DE BARRA DE DOS EJES PARA 
INCLINACIÓN DEL ACELERÓMETRO  
Esta actividad desarrolla una gráfica de barra de dos ejes que es útil para desplegar 
gráficamente las mediciones de inclinación del acelerómetro. 

Partes y Equipo 

Use el circuito del Capítulo 3, actividad #2, page 71. 

Caracteres personalizados de Gráfica de Barra Vertical  

La subrutina Def_Vert_Bar_Char en el siguiente programa ejemplo define cualquiera 
de los 16 diferentes caracteres de barra vertical personalizados. La Figura 7-3 muestra 
como cada caracter de barra corresponde a un valor. 
 

 

Figura 7-3 
Valores de gráfica de 
barra vertical con 
Caracteres 

   
Las definiciones de caracteres personalizados (Figura 7-3) siguen una secuencia que se 
relaciona directamente a los valores que representan, entonces las definiciones pueden ser 
calculadas según se requiera en vez de ser guardadas en la EEPROM. Puesto que cada 
definición de caracter personalizado toma 8 bytes, eliminar 16 definiciones salvará 128 
bytes de memoria de programación.   
 
La subrutina Def_Vert_Bar_Char define los caracteres de la Figura 7-3. La variable 
value define cual de los 0 a 15 caracteres será desplegado. Cada bit en la variable 
rowMap determina si uno de los renglones en el caracter personalizado es blanco o negro.   

 
Def_Vert_Bar_Char: 
 
  SEROUT LcdPin, T9600, 
         [Define + custChar] 

  rowMap = %1111111100000000 >> (value & %1111) 
 
  FOR index = 0 TO 7                          
    IF rowMap.LOWBIT(index) = 1 THEN          
      SEROUT LcdPin, T9600, [%11111]                 
    ELSE 
      SEROUT LcdPin, T9600, [%00000]              
    ENDIF 
  NEXT 
 
  RETURN 
 

El comando rowMap = %1111111100000000 >> (value & %1111) recorre los ocho 
bits 1 de %1111111100000000 hacia la derecha tantos lugares como lo indique la 
variable value.  El término (value & %1111) es llamado una mascara de bits y hace 
posible usar valores de 16 a 31 para darle los mismos resultados que de 0 a 15.  Si value 
es 3, el comando rowMap = %1111111100000000 >> (value & %1111) guarda 
%0001111111100000 en la variable rowMap.  Puesto que rowMap es una variable de solo 
un byte, solo guarda los 8 bits más bajos del término, o sea %11100000.  Comparelo con 
el caracter personalizado para 3 en la Figura 7-3.  Si value es 4, el resultado del byte 
bajo de la variable rowMap es %11110000. Ahora, observe el caracter 4 en la Figura 7-3.  
Inténtelo para cada valor (0 al 15) y verá que el patron de bits en el byte rowMap coincide 
con el patrón de los renglones con pixeles negros en la Figura 7-3. 
 
Definir un caracter personalizado involucra enviar 8 bytes. Los valores binarios en cada 
byte sucesivo define cada uno de los 8 renglones en el caracter, de arriba a abajo. El ciclo 
FOR...NEXT en la subrutina Def_Vert_Bar_Char tiene un bloque de código IF...THEN que 
usa el operator.LOWBIT para revisar cada bit en la variable rowMap y lo usa para definir 
un renglón de 5 pixeles blancos o negros en el caracter personalizado.  Digamos que 
value es 3, entonces rowMap es %11100000. La primera vez en el ciclo FOR...NEXT, 
index es 0 y entonces la línea IF...THEN examina rowMap.LOWBIT(0), el bit más a la 
derecha. Puesto que es 0, la línea IF...THEN manda un byte que contiene %00000 a la 
LCD, que hace que todos los pixeles sean blancos. La sexta vez en el  ciclo FOR...NEXT, 
rowMap.LOWBIT(5) = 1, entonces la línea IF...THEN manda un comando SEROUT con un 
byte conteniendo %11111.  Entonces, los renglones sexto, séptimo y octavo serán negros, 
que resulta en el caracter mostrado en la Figura 7-3. 
 

 
SEROUT 14, 84, [%11111] de hecho manda el byte %00011111, y los 5 bits más 
bajos son los que usa la LCD para definir sus renglones de 5 pixeles de ancho.   

Programa Ejemplo: TestVerticalBars.bs2 

Este programa ejemplo despliega los 16 diferentes caracteres personalizados de la barra 
vertical en la Línea 0, carácter 8 de la LCD Parallax. Lo hace en rápida sucesión, una y 
otra vez. También despliega el número que value guarda en la Terminal de Depuración.   
 

√ Abra TestVerticalBars.bs2 en el Editor del BASIC Stamp. 
√ Córralo y verifique que todos los caracteres de la Figura 7-3 son desplegados. 
√ Incremente el argumento  Duration del  comando PAUSE. 
√ Vuelva a correr el programa y verifque que las barras en la LCD representan 

correctamente los números desplegados en la Terminal de Depuración. 
 
' -----[ Titulo ]------------------------------------------------------------- 
' Sensores Inteligentes y sus Aplicaciones - TestVerticalBars.bs2 
' Despliega 16 caracteres de barra vertical diferentes en Linea 0, caracter-8 
' de la LCD Serial Parallax. 
 
' {$STAMP BS2}                       ' Dispositivo Objeto  = BASIC Stamp 2 
' {$PBASIC 2.5}                      ' Lenguaje            = PBASIC 2.5 
 
' -----[ Pins E/S]----------------------------------------------------------- 
 
LcdPin         PIN     14                   ' pin E/S conectado a RX de la LCD  
 
' -----[ Constants ]---------------------------------------------------------- 
 
T9600          CON     84                    ' True, 8-bits, no parity, 9600 
 
LcdCls         CON     12                    ' Form feed -> limpia pantalla 
LcdOn          CON     22                    ' Enciende la pantalla 
Line0          CON     128                   ' Linea 0, caracter 0 
Line1          CON     148                   ' Linea 1, caracter 0 
Define         CON     248                   ' direccion define caract pers 0 
 
' -----[ Variables ]---------------------------------------------------------- 
 
custChar       VAR     Byte                ' selector de carcter personalizado 
index          VAR     Nib                   ' variable indice Eeprom  
rowMap         VAR     Byte                  ' linea punteada de 5-pixel 
cursor         VAR     Byte                  ' Colocacion de Cursor  
value          VAR     Byte                  ' Valor a ser graficado 
line           VAR     Byte                  ' Linea0 o Linea1 
 
' -----[ Inicialización ]----------------------------------------------------- 
 
' LCD initializiation. 
PAUSE 100                                    ' Estabiliza fuente de poder 

SEROUT LcdPin, T9600, [LcdOn, LcdCls]        ' Inicializa LCD 
PAUSE 5                                   ' 5 ms retraso para limpiar pantalla 
 
' subrutina valores Caracter personalizado. 
custChar = 3                                 ' Use caracter personalizado 3 
line = Line0                                 ' Cursor a Linea0 
cursor = 7                                   ' Cursor a 8avo caracter 
 
' -----[ Rutina Principal ]--------------------------------------------------- 
 
DO                                           ' Ciclo Principal 
 
  FOR value = 0 TO 16                        ' value cuenta de 0 a 16 
    DEBUG ? value                    ' Muestra value en Terminal de Depuración 
    GOSUB Def_Vert_Bar_Char              ' Define caracter de grafica de barra  
    GOSUB Disp_Cust_Char                     ' Muestra caracter en la LCD 
    PAUSE 50                                 ' Hace lento al ciclo 
  NEXT 
 
  DEBUG CR, CR                               ' espacios antes sig. secuencia 
  PAUSE 500                                  ' Retraso antes sig. secuencia 
 
LOOP                                         ' Repite ciclo principal 
 
' -----[ Subrutina - Def_Vert_Bar_Char ]------------------------------------- 
 
' Define un caracter de grafica de barra vertical basado en la variable value  
' (0 to 15) y la variable custChar, que selecciona los caracteres  
' personalizados de la LCD Serial Parallax entre 0 y 7. 
 
Def_Vert_Bar_Char: 
  ' Inicia definicion de caracter personalizado 
  SEROUT LcdPin, T9600, 
         [Define + custChar] 
 
  ' Selecciona un mapa de renglon para el caracter personalizado segun value. 
  rowMap = %1111111100000000 >> (value & %1111) 
 
  ' Manda 8 bytes, cada uno definiendo un renglón punteado en el caracter  
  ' personalizado. Cada renglon es definido examinando un bit en la variable  
  ' rowMap y luego enviando %11111 si el bit es 1 o %00000 si el bit es 0. 
  FOR index = 0 TO 7                         ' Repite 7 veces, index cuenta 
    IF rowMap.LOWBIT(index) = 1 THEN         ' Examina siguiente bit en rowMap 
      SEROUT LcdPin, T9600, [%11111]         ' Si 1, envia %11111 
    ELSE 
      SEROUT LcdPin, T9600, [%00000]         ' De lo contrario, envia %00000 
    ENDIF 
  NEXT 
                                             ' Regresa de la subrutina. 
  RETURN 

' -----[ Subrutinas - Disp_Cust_Char ]--------------------------------------- 
 
' Esta subrutina despliega un caracter personalizado. La variable line puede 
' ser establecida en Line0 o Line1 y la variable cursor puede ser establecida 
' en un valor entre 0 y 15. La variable custChar selecciona uno de los 
' caracteres personalizados de la LCD y debe estar en un valor entre 0 y 7. 
 
Disp_Cust_Char: 
  SEROUT LcdPin, T9600,                    ' Escribe caracter personalizado 
         [line + cursor, custChar] 
  RETURN 
 

Su Turno - 31 Niveles cubriendo 2 renglones y Agregando un eje Horizontal 

El comando que establece el patron de bits en la variable rowMap fué 
rowMap = %1111111100000000 >> (value & %1111). El cálculo value & %1111 
resultará 0 si value es 16, en 1 si value es 17, etc., hasta 15 si value es 31. Este patron 
continuará sin importer que tan grande es value. Puesto que la gráfica de barra se 
comportará de la misma forma para valores entre 16 y 31 como lo hace para valores entre 
0 y 15, la gráfica de barra puede ser colocada en la línea superior o en la inferior para 
indicar en qué rango cae la variable value.  El rango real de desplegado es de 1 a 31, con 
el 16 mostrando ninguna barra.   
 
Modificar el programa para mostrar los caracteres en esta forma involucra solo unos 
pocos cambios a la Rutina Principal. Primero, los argumentos StartValue y EndValue 
del ciclo FOR...NEXT tienen que ser cambiados de (0 a 15) a (1 a 31). También, un 
comando SEROUT tiene que ser agregado para sobreescribir los  caracteres personalizados 
previos con espacios en blanco. Luego, puede ser agregado un bloque de código 
IF...THEN...ELSE que posicione el cursor en la Linea 0 o la Linea 1. Debe colocar el 
cursor en la Linea 0 si value es mayor a 16, o en la línea inferior si es menor o igual a 16. 
 

√ Salve TestVerticalBars.bs2 como TestVerticalBarsYourTurn.bs2. 
√ Modifique el DO...LOOP en la rutina principal como se muestra a continuación. 
√ Corra el programa y verifique que la gráfica de barra ahora despliega 31 niveles 

diferentes, del 1 al 31. 
 
DO 
 
  FOR value = 1 TO 31 
    SEROUT 14, 84, [Line0 + cursor, " ", 
                    Line1 + cursor, " "] 
    IF value <= 16 THEN  

      line = Line1  
    ELSE  
      line = Line0 
    ENDIF 
    DEBUG ? value                          
    GOSUB Def_Vert_Bar_Char                   
    GOSUB Disp_Cust_Char                      
    PAUSE 50                                  
  NEXT 
 
  DEBUG CR, CR                                
  PAUSE 500                                   
 
LOOP       
 

Puede también anidar el ciclo FOR value = 1 TO 31...NEXT dentro de un ciclo FOR 
cursor = 0 TO 15...NEXT y causar que la gráfica de barra se mueva a través de la 
pantalla cada vez que se repite.  Controlando asi la compensación del cursor, la gráfica de 
barra puede desplegar 2 ejes con valores verticales de 1 a 31 y valores horizontales de 0 a 
15. 
 

√ ¡Inténtelo!                                    

Una pantalla de dos ejes 

Si bien la anterior sección Su Turno demostró como desplegar caracteres en 2 ejes, una 
pantalla completamente en blanco no comunica realmente que el valor está a la mitad de 
su rango. Una mejor forma de hacer llegar el mensaje es haciendo que aparezcan 2 
caracteres personalizados uno junto al otro. La Figura 7-4 muestra cómo trabaja esto. Los 
primeros 2 ejemplos de pantalla no están a la mitad ni del rango horizontal ni del vertical 
y entonces se despliegan caracteres simples. El tercer ejemplo está a la mitad del rango 
horizontal (8), y entonces la medición vertical se despliega con dos caracteres adyacentes. 
El cuarto ejemplo muestra que la medición está a la mitad de su rango vertical, y 
entonces los renglones en el carácter de arriba y abajo del valor tienen pixeles negros.  
Cuando the medición está centrada en ambas la horizontal y la vertical el quinto ejemplo 
muestra cómo se debería ver. Estas características son especialmente importantes para 
graficamente indicar cuando uno o los dos ejes del acelerómetro están nivelados. 

Figura 7-4 
Puntos trazados en la 
gráfica de barra de dos 
ejes  

 
Si bien las características mostradas en la Figura 7-4 hacen que la pantalla esté un poco 
menos cifrada, el programa tiene que tomar muchas más decisiones. En el siguiente 
programa ejemplo, las subrutinas Bar_Graph_H_V y Horizontal_Placement se encarga 
de todas las decisiones que apoyan estas características adicionales.   

Programa Ejemplo: TwoAxisBarDisplay.bs2 

 

Descarga gratis Este programa está disponible como descarga grátis .bs2 en la página del 
producto de Sensores Inteligentes y sus Aplicaciones en www.parallax.com. 

 
Puede probar este programa ejemplo introduciendo coordenadas horizontales y verticales 
en la Ventana de Transmisión de la Terminal de Depuración. 
 

√ Abra y corra TwoAxisBarDisplay.bs2 con the Editor del BASIC Stamp. 
√ Introduzca valores en el rango de 0 a 16 para valores horizontales y de 1 a 31 

para verticales, y observe los resultados desplegados por la LCD.   

√ Intente los valores de la Figura 7-4 y verifique que concuerdan con las pantallas. 
√ Pruebe los valores horizontales 0, 8, y 16 para varios valores verticales. 
√ Pruebe los valores verticales 1, 16, y 31 para varios valores horizontales. 

 
Otra forma de probar la pantalla es con la Rutina principal alterna mostrada a 
continuación. Tiene ciclos FOR...NEXT anidados que prueban toda la variable value en 
todos los valores posibles de la variable cursor. 
 

√ Salve una copia de TwoAxisBarDisplay.bs2 como TwoAxisBarDisplayTest.bs2. 
√ Corra el programa modificado y observe para asegurarse que todos los valores 

son desplegados correctamente. 
 
' -----[ Rutina Principal ]-------------------------------------------- 
 
DO   
  FOR cursor = 0 TO 16 
    FOR value = 1 TO 31 
      GOSUB Bar_Graph_H_V 
      PAUSE 200 
    NEXT 
    SEROUT 14, 84, [LcdCls] 
    PAUSE 500 
  NEXT 
LOOP 
 

' -----[ Titulo ]------------------------------------------------------------- 
' Sensores Inteligentes y sus Aplicaciones - TwoAxisBarDisplay.bs2 
' Despliega un caracter que cambia vertical y horizontalmente. 
 
' {$STAMP BS2}                           ' Dispositivo Objeto  = BASIC Stamp 2 
' {$PBASIC 2.5}                          ' Lenguaje            = PBASIC 2.5 
 
' -----[ Pins E/S]----------------------------------------------------------- 
 
LcdPin         PIN     14                   ' pin E/S conectado a RX de la LCD 
 
' -----[ Constants ]---------------------------------------------------------- 
 
T9600        CON     84                    ' True, 8-bits, no parity, 9600 
 
LcdCls         CON     12                    ' Form feed -> limpia pantalla 
LcdCr          CON     13                    ' Retorno de carro  
LcdOff         CON     21                    ' Apaga la pantalla 
LcdOn          CON     22                    ' Enciende la pantalla 
Line0          CON     128                   ' Linea 0, caracter 0 
Line1          CON     148                   ' Linea 1, caracter 0 
Define         CON     248                   ' direccion define caract pers 0 

' -----[ Variables ]---------------------------------------------------------- 
 
custChar       VAR     Byte               ' selector de carcter personalizado 
index          VAR     Nib                   ' variable indice Eeprom 
rowMap         VAR     Byte                  ' linea punteada de 5-pixel 
cursor         VAR     Byte                  ' Colocacion de Cursor 
value          VAR     Byte                  ' Valor a ser graficado 
line           VAR     Byte                  ' Linea0 or Linea1 
 
' -----[ Inicialización ]----------------------------------------------------- 
 
PAUSE 100                                    ' Estabiliza fuente de poder 
SEROUT LcdPin, T9600, [LcdOn, LcdCls]        ' Inicializa LCD 
PAUSE 5                                   ' 5 ms retraso para limpiar pantalla 
 
' -----[ Rutina Principal ]--------------------------------------------------- 
 
DO                                           ' Ciclo Principal 
 
  DEBUG "Enter horizontal",                  ' Pide posicion de caracter 
        CR, "value (0 TO 16)" 
  DEBUGIN DEC cursor                         ' Obtiene posicion de caracter 
  DEBUG "Enter vertical", 
        CR, "value (1 TO 31)"                ' Pide al usuario valor 
  DEBUGIN DEC value                          ' Obtiene valor 
  GOSUB Bar_Graph_H_V                       ' Llama a Bar_Graph_H_V 
 
LOOP                                         ' Repite ciclo principal 
 
' -----[ Subrutinas - Bar_Graph_H_V ]--------------------------------------- 
 
' Define y despliega caracteres de gráfica de barra de 2 ejes segun valor de 
' cursor (0 to 16) y value (1 to 31). Llama a Def_Vert_Bar_Char y a 
' Horizontal_Placement. 
 
Bar_Graph_H_V: 
 
  SEROUT 14, 84, [LcdCls]                 ' Limpia trazado previo 
  PAUSE 5                                 ' Retraso 5 ms para limpiar pantalla 
 
  ' Decide whether to display on Line 0 or Line 1. 
  IF value >= 16 THEN line = Line0 ELSE Line = Line1 
 
  GOSUB Def_Vert_Bar_Char                    ' Define caracter personalizado 
  GOSUB Horizontal_Placement 
 
  IF value = 16 THEN                         ' Caso especial: value = 16 
 
    value = 1                                ' Linea 0 display 
    custChar = 2 

    GOSUB Def_Vert_Bar_Char 
    line = Line0 
    GOSUB Horizontal_Placement 
 
    value = 15                               ' Linea 1 Display 
    custChar = 3 
    GOSUB Def_Vert_Bar_Char 
    line = Line1 
    GOSUB Horizontal_Placement 
 
    value = 16                               ' Restaura value 
 
  ENDIF 
 
  RETURN 
 
' -----[ Subroutine - Def_Vert_Bar_Char ]------------------------------------- 
 
' Define un caracter de gráfica vertical de barra segun la variable value (0 a 
' 15) y la variable custChar, la cual selecciona los caracteres personalizados 
' entre 0 y 7 de la LCD Serial Parallax. 
 
Def_Vert_Bar_Char: 
 
  ' Inicia definicion de caracter personalizado 
  SEROUT LcdPin, T9600, 
         [Define + custChar] 
 
  ' Selecciona un mapa de renglon para el caracter personalizado según value. 
  rowMap = %1111111100000000 >> (value & %1111) 
 
  ' Manda 8 bytes, cada uno definiendo un renglón punteado en el caracter  
  ' personalizado. Cada renglon es definido examinando un bit en la variable  
  ' rowMap y luego enviando %11111 si el bit es 1 o %00000 si el bit es 0. 
  FOR index = 0 TO 7                         ' Repite 7 veces, index cuenta 
    IF rowMap.LOWBIT(index) = 1 THEN         ' Examina siguiente bit en rowMap 
      SEROUT 14, 84, [%11111]                ' Si 1, envia %11111 
    ELSE 
      SEROUT 14, 84, [%00000]                ' De lo contrario, envia %00000 
    ENDIF 
  NEXT 
                                             ' Regresa de la subrutina. 
  RETURN 
 
' -----[ Subrutinas - Horizontal_Placement ]--------------------------------- 
 
' Coloca la gráfica de barra vertical en una de las 16 columnas verticales de  
' la LCD 2X16 de Parallax. La variable cursor puede poner la posicion 
' horizontal en valores entre 0 y 16 (centro en 8).  Llama a Disp_Custom_Char. 
 
Horizontal_Placement: 

  SELECT cursor                              ' Cursor 0 a 7, sin cambios 
    CASE 0 TO 7 
      GOSUB Disp_Cust_Char 
    CASE 8                                   ' Cursor 8, despliega en 7 & 8 
      cursor = 7 
      GOSUB Disp_Cust_Char 
      cursor = 8 
      GOSUB Disp_Cust_Char 
    CASE 9 TO 16                             ' Cursor 9 a 16, despliega 1 izq 
      cursor = cursor - 1 
      GOSUB Disp_Cust_Char 
      cursor = cursor + 1 
    ENDSELECT 
 
  RETURN 
 
' -----[ Subrutinas - Disp_Cust_Char ]--------------------------------------- 
 
' Esta subrutina despliega un caracter personalizado. La variable line puede 
' ser establecida en Line0 o Line1 y la variable cursor puede ser establecida 
' en un valor entre 0 y 15. La variable custChar selecciona uno de los 
' caracteres personalizados de la LCD y debe estar en un valor entre 0 y 7. 
 
Disp_Cust_Char: 
 
  SEROUT LcdPin, T9600,                    ' Escribe caracter personalizado 
         [line + cursor, custChar] 
  RETURN 
 

Su Turno - Desplegado Gráfico de inclinación en dos ejes  

Indicar la inclinación con TwoAxisBarDisplay.bs2 involucra comandos PULSIN para 
obtener las mediciones del acelerómetro en los ejes x y y.  También require escalar y 
compensar para ajustar las mediciones del acelerómetro en una escala vertical de 31 y 
una escala horizontal de 17. 
 
La escala horizontal también tiene que ser invertida. La Figura 7-5 muestra cómo los ejes 
x y y del acelerómetro se relacionan con los ejes horizontal y vertical de la LCD.  Note 
que la dirección positiva del eje y apunta a la dirección contraria en la que los valores del 
eje horizontal de la LCD se incrementan. Cuando el valor escalado del eje y sea 16, la 
pantalla debiera mostrar 0 y cuando el valor escalado del eje y sea 0, la pantalla debiera 
mostrar 16.  La manera de corregir esto es usar un comando que reste a 16 el valor 
escalado del eje y. Esencialmente, esto es cursor = 16 − (valor escalado de y). 

Figura 7-5: Ejes x y y del acelerómetro  
 

 
 
La forma más fácil de convertir TwoAxisBarDisplay.bs2 en una herramienta de 
nivelación bi-axial es tomar prestados los bloques de código del Capítulo 3 y ajustarlos 
según se requiera. Las variables de los ejes x y y, comandos PULSIN y los comandos 
para escalamiento y compensación de HorizontalTilt.bs2 en el Capítulo 3, actividad #6 
proveen un punto de inicio.  Los comandos PULSIN y escalamiento pueden reemplazar 
los comandos DEBUG y DEBUGIN en la Rutina Principal de TwoAxisBarDisplay.bs2. La 
subrutina Bar_Graph_H_V usa las variables value y cursor para graficar en la LCD.  
Antes de que pueda ser llamada, value necesita ser igualada al valor escalado del eje x y 
cursor necesita ser igualada al valor escalado del eje y, restado a 16.   
 

 
Obtener las constantes de escalamiento ** se dejan como ejercicios al final del ejercicio. 

 
 

√ Salve TwoAxisBarDisplay.bs2 como TwoAxisBarTiltDisplay.bs2. 
√ Agregue estas 2 declaraciones a la sección de Variables del programa. 

 
X     VAR     Word                  ' eje x del Acelerómetro  
y     VAR     Word                  ' eje y del Acelerómetro  
 

√ Reemplace el DO...LOOP más externo en la rutina principal con: 

DO                                           ' Ciclo Principal 
 
  PULSIN 6, 1, x 
  PULSIN 7, 1, y 
 
  value = (x MIN 1875 MAX 3125) - 1875 ** 1625 + 1 
  cursor = 16 - ((y MIN 1875 MAX 3125) - 1875 ** 891) 
 
  GOSUB Bar_Graph_H_V                       ' Llama a Bar_Graph_H_V 
 
LOOP                                        ' Repite ciclo principal  
 

√ Salve y corra el programa y pruébelo bajo una variedad de circunstancias de 
inclinación. 

RESUMEN 
La mediciones de distancia e inclinación se prestan para desplegarlas en gráficas de barra.  
La mayoría de las gráficas de barra pueden hacerse usando solo uno o dos de los 8 
espacios de memoria para caracteres personalizados de la LCD Serial Parallax. Las 
definiciones de caracter personalizado para las gráficas de barra pueden ser guardadas en 
la  EEPROM del BASIC Stamp, o definidas por código que relacione el valor que está 
siendo graficado al número de lineas o columnas de pixeles que necesitan ser 
desplegados.    
 
Guardar caracteres personalizados en la EEPROM involucra directivas DATA que 
contengan 8 bytes cada una.  Los 5 bits más bajos de cada byte guardan el patron de bits 
para cada una de las 8 lineas (de 5 pixeles de ancho) que componen un caracter.  
Nombres Símbolo precediendo directivas DATA hacen posible copiar la dirección inicial 
de los bytes de una directiva DATA determinada en una variable. Una subrutina puede usar 
la variable que guarda la dirección de inicio para operaciones READ que secuencialmente 
copia bytes a una variable, luego envía el contenido de esa variable a la LCD. 
 
Desplegar gráficas de barra es típicamente mas  eficiente con la memoria si los caracteres 
pueden ser generados en base al valor. Cuando la gráfica de barra es horizontal, un 
comando LOOKUP es útil para guardar valores binarios que correspondan con el número 
de columnas que tendrán pixeles negros en un caracter personalizado determinado. Para 
gráficas verticales, los bits pueden ser cambiados en una variable.  En cualquier caso, las 
subrutinas pueden examinar los bits en una variable, luego oscurecer pixeles en un cierto 
número of renglones o columnas en un caracter.     

Preguntas 

1. ¿Qué parte de las directivas DATA hace posible encontrar la dirección de inicio de 
una definición de caracter?  

2. ¿Cómo le dicen los programas en este capítulo a la subrutina Def_Cust_Char 
qué caracter definir y dónde encontrar la definición de caracter? ¿Cómo 
especifica una posición de caracter para la subrutina Disp_Cust_Char? Si 
quiere desplegar 2 caracteres personalizados diferentes al mismo tiempo, ¿puede 
usar la misma definición de caracter personalizado? ¿Porqué? 

3. ¿Cómo guarda información HorizBarGraph.bs2 acerca de qué columnas en un 
caracter personalizado deben tener pixeles blancos o negros? 

4. ¿Porqué puede usar solo un byte en vez de 8 para definir un caracter 
personalizado con ciertas columnas negras? 

5. ¿Qué es lo que hace la instrucción IF rowMap.LOWBIT(index) = 1 THEN... do 
en TestVerticalBars.bs2? 

6. ¿Porqué puede usar solo un byte en vez de 8 para definir un caracter 
personalizado con ciertas columnas negras? 

7. ¿Qué es lo que hace la instrucción IF rowMap.LOWBIT(index) = 1 THEN... do 
en el programa TestVerticalBars.bs2?   

Ejercicios 

1. Escriba una directiva DATA para caracter personalizado para un caracter con 
todas las líneas horizontales. 

2. Derive el valor 1625 en el comando value = (x MIN 1875 MAX 3125) - 
1875 ** 1625 + 1. 

Proyectos 

1. Modifique PingBarGraph.bs2 para que despliegue la medición de distancia en 
centímetros en la línea inferior. 

2. Modifique TwoAxisBarTiltDisplay.bs2 para que despliegue barras que 
correspondan con el ángulo de inclinación. Recuerde que ese ángulo es una 
función de la gravedad sensada por cada uno de los ejes del acelerómetro. Tip: 
Use el código de ejemplo del Capítulo 3, actividad #6 para cálculos del ángulo 
de inclinación. 

Soluciones 

Q1. El nombre Símbolo. 
Q2. Los programas tienen que establecer 2 variables antes de hacer la llamada a 

la  subrutina Def_Cust_Char: custChar y charBase.  La variable 
custChar debe guardar el valor que el programa usará para decirle a la 
LCD que despliegue el caracter personalizado.  Las definiciones de 
caracteres personalizados de la LCD pueden ser un valor de 0 a 7 y 
entonces custChar tiene que caer en ese rango. El nombre Símbolo en la 
directiva DATA del caracter personalizado tiene que ser guardada en la 
variable charBase. Puesto que variable = SymbolName guarda la 
dirección EEPROM del primer byte en una directiva DATA definida por 
SymbolName como variable, hace posible pasar la dirección de inicio de una 
directiva DATA de un caracter personalizado a la subrutina 
Def_Cust_Char. 

Q3. Establece el valor de las variables cursor y line antes de  llamar a la 
subrutina.  The rutina Disp_Cust_Char posicionara al cursor usando estas 
2 variables antes de decirle a la LCD que despliegue el caracter 
personalizado en esa posición. 

Q4. No, porque todas las instancias del mismo caracter personalizado 
actualmente desplegado por la LCD será actualizado tan pronto como la 
definición de caracter sea actualizada. 

Q5. La variable dotLine termina guardando uno de 5 valores: %00000, 
%10000, %11000, %11100, o %11110. Cada dígito corresponde a una de 
las 5 columnas verticales de pixeles en un caracter personalizado. Si el 
valor es 1, los pixeles en esa columna serán puestos en negro.  Si es 0, serán 
dejados en blanco. 

Q6. Cada uno de los 8 renglones en la definición del caracter personalizado 
tendrá el mismo patrón, entonces el programa solo tiene que enviar el 
mismo byte de renglón 8 veces para definir cada renglón del caracter 
personalizado.   

Q7. Cuando la condición IF…THEN evalúa a 1, los pixeles para el renglón se 
establecen en negro con SEROUT 14, 84, [%11111]; de otra forma, se 
establecen en blanco con SEROUT 14, 84, [%00000].  

E1. Ejemplo de solución: 

Stripes        DATA        %11111,           ' * * * * * 
                           %00000,           '  
                           %11111,           ' * * * * * 
                           %00000,           '  
                           %11111,           ' * * * * * 
                           %00000,           '  
                           %11111,           ' * * * * * 
                           %00000            '  
 

E2. El objetivo de la expresión value = (x MIN 1875 MAX 3125) - 1875 ** 
1625 + 1 en la actividad #3 es escalar el valor del eje x del acelerómetro, que 
debe estar en algún punto entre 1875 y 3125, a un valor entre 1 a 31. Esto eligirá 
una de las 31 posibles variantes de gráfica de barra vertical para la subrutina 
Bar_Graph_H_V. Para cuando el operator ** es alcanzado, 1875 ha sido restado, 
entonces el rango es 0 a 1250 (1251 elementos en la escala de entrada), y 
queremos escalarlo al rango de 1 a 31 (31 elementos en la escala de salida).  Del 
Capítulo 3, actividad #3, sabemos que ** ScaleConstant = Int[65536 × 
(elementos en la escala de salida ÷ (elementos en la escala de entrada – 1))].  
Entonces, Int[65536 × (31 ÷ (1251 – 1))] = Int[65536 × (31 ÷ 1250)] = Int[65536 
× 0.0248] = Int[1625.2928] = 1625. 

 
P1. Solution Ejemplo– Rutina principal modificada de PingBarGraph.bs2. 

 
' -----[ Rutina Principal ]-------------------------------------------- 
 
DO                                    ' Ciclo Principal 
 
  PULSOUT 15, 5 
  PULSIN 15, 1, time 
 
  cmDistance = CmConstant ** time 
 
  DEBUG HOME, DEC3 cmDistance, " cm" 
 
  value = cmDistance 
 
  GOSUB Bar_Graph                     ' Despliega como gráfica de barra 
 
  SEROUT 14, 84, [Line1, DEC4 value,  ' Despliega medición cm  
                  " cm"]               ' en Linea 1  
 
  PAUSE 100 
 
LOOP                                  ' Repite ciclo principal 

P2. Solution Ejemplo– Vea los comentarios en la sección Titulo par aver cómo fue 
construído este programa a partir de programas ejemplo previos y cálculos 
explicados en el texto: 

 
' -----[ Titulo ]------------------------------------------------------------- 
' Sensores Inteligentes y sus Aplicaciones - Ch7Proj2.bs2 
' Despliega un caracter que cambia vertical y horizontalmente según el ángulo 
' de inclinación del MX2125 en grados. Este programa es una combinacion de 
' HorizontalTilt.bs2 del Capítulo 3, actividad #6 y TwoAxisBarTiltDisplay.bs2 
' del Capítulo 7, actividad #3. La medición del eje x tiene que ser cambiada 
' de -90 a +90 grados a un valor de 1 a 31 usando la equation de constante de  
' escalamiento **. De igual forma, la medición angular del eje y de -90 a +90 
' tiene que ser cambiada a un valor de 0 a 16. Los calculos con comentados y 
' mostrados en la rutina principal justo antes su uso con los operadores **. 
 
' {$STAMP BS2}                           ' Dispositivo Objeto  = BASIC Stamp 2 
' {$PBASIC 2.5}                          ' Lenguaje            = PBASIC 2.5 
 
' -----[ Pines E/S]----------------------------------------------------------- 
 
LcdPin         PIN     14                   ' pin E/S conectado a RX de la LCD 
 
' -----[ Constants ]---------------------------------------------------------- 
 
T9600        CON     84                    ' True, 8-bits, no parity, 9600 
 
LcdCls         CON     12                    ' Form feed -> limpia pantalla 
LcdCr          CON     13                    ' Retorno de carro 
LcdOff         CON     21                    ' Apaga la pantalla 
LcdOn          CON     22                    ' Enciende la pantalla 
Line0          CON     128                   ' Linea 0, caracter 0 
Line1          CON     148                   ' Linea 1, caracter 0 
Define         CON     248                   ' direccion define caract pers 0 
Negative       CON     1              ' Signo - .bit15 de variables de palabra  
Positive       CON     0 
 
' -----[ Variables ]---------------------------------------------------------- 
 
custChar       VAR     Byte                ' selector de carcter personalizado 
index          VAR     Nib                   ' variable indice Eeprom 
rowMap         VAR     Byte                  ' linea punteada de 5-pixel 
cursor         VAR     Byte                  ' Colocacion de Cursor 
value          VAR     Byte                  ' Valor a ser graficado 
line           VAR     Byte                  ' Linea0 or Linea1 
x              VAR     Word 
y              VAR     Word 
sine           VAR     Word             ' seno en circulo r = 127 
side           VAR     Word             ' variable para disparo de subrutinas  
angle          VAR     Word             ' resultado angulo - grados 
sign           VAR     Bit              ' bit de signo  

' -----[ inicialización ]----------------------------------------------------- 
 
PAUSE 100                                    ' Estabiliza fuente de poder 
SEROUT LcdPin, T9600, [LcdOn, LcdCls]        ' Inicializa LCD 
PAUSE 5                                   ' 5 ms retraso para limpiar pantalla 
' -----[ Main Rutina ]------------------------------------------------------- 
 
DO                                           ' Ciclo Principal 
 
  PULSIN 6, 1, x 
  PULSIN 7, 1, y 
 
  x = (x MIN 1875 MAX 3125) - 1875 ** 13369 - 127 
  y = (y MIN 1875 MAX 3125) - 1875 ** 13369 - 127 
 
  side = x 
  GOSUB Arcsine 
  ' Int[65536 * (31 / (181 - 1))] = 11286 
  value = angle + 90 ** 11286 + 1 
 
  side = y 
  GOSUB Arcsine 
  ' Int[65536 * (17 / (181 - 1))] = 6189 
  cursor = 16 - (angle + 90 ** 6189) 
 
 
  GOSUB Bar_Graph_H_V                        ' Llama Bar_Graph_H_V 
 
LOOP                                         ' Repite ciclo principal 
 
' -----[ Subrutinas - Bar_Graph_H_V ]--------------------------------------- 
 
' Define y despliega caracteres de gráfica de barra de 2 ejes según el valor 
' de cursor (0 to 16) y value (1 to 31). Llama Def_Vert_Bar_Char y 
' Horizontal_Placement. 
 
Bar_Graph_H_V: 
 
  SEROUT 14, 84, [LcdCls]                 ' Limpia trazado previo 
  PAUSE 5                                 ' Retraso 5 ms para limpiar pantalla 
 
  ' Decide si escribe en la Linea 0 o en la Linea 1. 
  IF value >= 16 THEN line = Line0 ELSE Line = Line1 
 
  GOSUB Def_Vert_Bar_Char                    ' Define caracter personalizado 
  GOSUB Horizontal_Placement 
 
  IF value = 16 THEN                         ' Caso especial: value = 16 

    value = 1                                ' Despiega Linea 0  
    custChar = 2 
    GOSUB Def_Vert_Bar_Char 
    line = Line0 
    GOSUB Horizontal_Placement 
 
    value = 15                               ' Despiega Linea 1  
    custChar = 3 
    GOSUB Def_Vert_Bar_Char 
    line = Line1 
    GOSUB Horizontal_Placement 
 
    value = 16                               ' Restaura value 
 
  ENDIF 
 
  RETURN 
 
' -----[ Subrutina - Def_Vert_Bar_Char ]------------------------------------- 
 
' Define un caracter de gráfica vertical de barra segun la variable value (0 a 
' 15) y la variable custChar, la cual selecciona los caracteres personalizados 
' entre 0 y 7 de la LCD Serial Parallax. 
 
Def_Vert_Bar_Char: 
 
  ' Inicia definicion de caracter personalizado 
  SEROUT LcdPin, T9600, 
         [Define + custChar] 
 
  ' Selecciona un mapa de renglon para el caracter personalizado según value. 
  rowMap = %1111111100000000 >> (value & %1111) 
 
  ' Manda 8 bytes, cada uno definiendo un renglón punteado en el caracter  
  ' personalizado. Cada renglon es definido examinando un bit en la variable  
  ' rowMap y luego enviando %11111 si el bit es 1 o %00000 si el bit es 0. 
  FOR index = 0 TO 7                         ' Repite 7 veces, index cuenta 
    IF rowMap.LOWBIT(index) = 1 THEN         ' Examina siguiente bit en rowMap 
      SEROUT 14, 84, [%11111]                ' Si 1, envia %11111 
    ELSE 
      SEROUT 14, 84, [%00000]                ' De lo contrario, envia %00000 
    ENDIF 
  NEXT 
                                             ' Regresa de la subrutina. 
  RETURN 
 
' -----[ Subrutinas - Horizontal_Placement ]--------------------------------- 
 
' Coloca la gráfica de barra vertical en una de las 16 columnas verticales de  
' la LCD 2X16 de Parallax. La variable cursor puede poner la posicion 
' horizontal en valores entre 0 y 16 (centro en 8).  Llama a Disp_Custom_Char. 

Horizontal_Placement: 
 
  SELECT cursor                              ' Cursor 0 a 7, sin cambios 
    CASE 0 TO 7 
      GOSUB Disp_Cust_Char 
    CASE 8                                   ' Cursor 8, despliega en 7 & 8 
      cursor = 7 
      GOSUB Disp_Cust_Char 
      cursor = 8 
      GOSUB Disp_Cust_Char 
    CASE 9 TO 16                             ' Cursor 9 a 16, despliega 1 izq 
      cursor = cursor - 1 
      GOSUB Disp_Cust_Char 
      cursor = cursor + 1 
    ENDSELECT 
 
  RETURN 
 
' -----[ Subrutinas - Disp_Cust_Char ]--------------------------------------- 
 
' Esta subrutina despliega un caracter personalizado. La variable line puede 
' ser establecida en Line0 o Line1 y la variable cursor puede ser establecida 
' en un valor entre 0 y 15. La variable custChar selecciona uno de los 
' caracteres personalizados de la LCD y debe estar en un valor entre 0 y 7. 
 
Disp_Cust_Char: 
 
  SEROUT LcdPin, T9600,                    ' Escribe caracter personalizado 
         [line + cursor, custChar] 
  RETURN 
 
' -----[ Subrutina - Arcsine ]----------------------------------------------- 
 
' Esta subrutina calcula el arcoseno segun la coordenada y en un circulo de 
' radio 127. Establezca la variable side igual a su coordenada y antes de  
' llamar a esta subrutina. 
 
Arcsine:                                ' Subrutina Seno Inverso  
  GOSUB Arccosine                       ' Obtiene coseno inverso  
  angle = 90 - angle                    ' sin(angle) = cos(90 - angle) 
  RETURN 
 
' -----[ Subrutina - Arccosine ]--------------------------------------------- 
 
' Esta subrutina calcula arcocoseno basado en la coordenada x en un circulo 
' de radio 127.  Establezca la variable side igual a su coordenada x antes 
' de llamar a esta subrutina. 
 
Arccosine:                                   ' Subrutina Inverso coseno 
  sign = side.BIT15                          ' Guarda el signo de side 

  side = ABS(side)                           ' Evalua side positivo 
  angle = 63 - (side / 2)                    ' aproximacion inicial al angulo 
  DO                                         ' Ciclo de approximation sucesiva 
    IF (COS angle <= side) THEN EXIT         ' Hecho cuando COS angle <= side 
    angle = angle + 1                        ' Mantiene incrementando angulo 
  LOOP 
  angle = angle */ 361                       ' Convierte brads a grados 
  IF sign = Negative THEN angle = 180 - angle' Ajusta si el signo es negativo. 
  RETURN 

Apéndice A: Tabla de caracteres ASCII 
 

Tabla ASCII (primeros 128 caracteres) 
Dec Hex Char Nombre / Función  Dec Hex Char  Dec Hex Char  Dec Hex Char 

0 00 NUL Null  32 20 space  64 40 @  96 60 ` 
1 01 SOH Start Of Heading  33 21 !  65 41 A  97 61 a 
2 02 STX Start Of Text  34 22 "  66 42 B  98 62 b 
3 03 ETX End Of Text  35 23 #  67 43 C  99 63 c 
4 04 EOT End Of Transmit  36 24 $  68 44 D  100 64 d 
5 05 ENQ Enquiry  37 25 %  69 45 E  101 65 e 
6 06 ACK Acknowledge  38 26 &  70 46 F  102 66 f 
7 07 BEL Bell  39 27 '  71 47 G  103 67 g 
8 08 BS Backspace  40 28 (  72 48 H  104 68 h 
9 09 HT Horizontal Tab  41 29 )  73 49 I  105 69 i 
10 0A LF Line Feed  42 2A *  74 4A J  106 6A j 
11 0B VT Vertical Tab  43 2B +  75 4B K  107 6B k 
12 0C FF Form Feed  44 2C ,  76 4C L  108 6C l 
13 0D CR Carriage Return  45 2D -  77 4D M  109 6D m 
14 0E SO Shift Out  46 2E .  78 4E N  110 6E n 
15 0F SI Shift In  47 2F /  79 4F O  111 6F o 
16 10 DLE Data Line Escape  48 30 0  80 50 P  112 70 p 
17 11 DC1 Device Control 1  49 31 1  81 51 Q  113 71 q 
18 12 DC2 Device Control 2  50 32 2  82 52 R  114 72 r 
19 13 DC3 Device Control 3  51 33 3  83 53 S  115 73 s 
20 14 DC4 Device Control 4  52 34 4  84 54 T  116 74 t 
21 15 NAK Non Acknowledge  53 35 5  85 55 U  117 75 u 
22 16 SYN Synchronous Idle  54 36 6  86 56 V  118 76 v 
23 17 ETB End Transmit Block  55 37 7  87 57 W  119 77 w 
24 18 CAN Cancel  56 38 8  88 58 X  120 78 x 
25 19 EM End Of Medium  57 39 9  89 59 Y  121 79 y 
26 1A SUB Substitute  58 3A :  90 5A Z  122 7A z 
27 1B ESC Escape  59 3B ;  91 5B [  123 7B { 
28 1C FS File Separator  60 3C <  92 5C \  124 7C | 
29 1D GS Group Separator  61 3D =  93 5D ]  125 7D } 
30 1E RS Record Separator  62 3E >  94 5E ^  126 7E ~ 
31 1F US Unit Separator  63 3F ?  95 5F _  127 7F delete 

 
Note que los códigos de control (los 32 caracteres ASCII más bajos) no tienen símbolos 
de pantalla estandarizados. Los caracteres listados para ellos son solo nombres usados en 
referencia a estos códigos.  Por ejemplo, para mover el cursor al principio de la siguiente 
línea de una impresora o terminal frecuentemente se requiere enviar los códigos line feed 
y carriage return. Este par es comúnmente referido como "LF/CR." 

Apéndice B: Documentación de la pantalla LCD 
Serial Parallax  
 

 

599 Menlo Drive, Suite 100 
Rocklin, California 95765, USA 
Office: (916) 624-8333 
Fax: (916) 624-8003 

 
General: info@parallax.com 
Técnico: support@parallax.com 
Web Site: www.parallax.com 
Educacional: www.stampsinclass.com 
 

Version 2.0 

LCD Serial Parallax  
2 renglones x 16 caracteres sin luz de fondo (#27976) 
2 renglones x 16 caracteres con luz de fondo (#27977) 
4 renglones x 20 caracteres con luz de fondo (#27979) 

 
Introducción 

Las pantallas LCDs Seriales Parallax son muy funcionales y de bajo costo y pueden ser 
fácilmente controladas por un microcontrolador BASIC Stamp®. Las LCD son de 2 o de 
4 renglones por 16 caracteres y proveen ajuste básico del texto para que su texto esté 
completo en pantalla. También le dan control completo sobre todas sus características 
avanzadas, permitiéndole mover el cursor a cualquier parte de la pantalla con una sola 
instrucción y apagarla y encenderla en cualquier configuración. Soportan los mismos 
caracteres visibles que la Terminal de Depuración del Editor de BASIC Stamp (ASCII 
Dec 32-127). Además, puede definir hasta 8 caracteres personalizados para desplegar en 
cualquier lugar de la LCD. 

Ideas de Aplicaciones  

¿Qué puede hacer una una LCD Serial Parallax? Si bien hay muchas posibilidades, esta 
es una pequeña lista de ideas con una LCD Serial y el BASIC Stamp de Parallax: 
 

• Una muy profesional interface de texto con el usuario en cualquier aplicación de 
microcontrolador   

• Una forma fácil de implementar la depuración de programas via serie sin una PC 
• Datos de salida en tiempo real en aplicaciones robóticas autónomas (Boe-Bot®, 

Toddler®, SumoBot®) 

Cables de Extensión para la LCD  

Las LCDs Seriales Parallax son compatible con nuestros cables Extension para LCD de 
14-pulgadas, parte #805-00012, en venta por separado en www.parallax.com.  Este cable 
de 3 pines hembra-hembra viene con un conector de  3 pines y fácilmente lo puede 
conectar su LCD a su tableta de proyectos. 

Código Muestra 

Se pueden buscar por número de parte en www.parallax.com: 
2x16 LCD Serial: búsqueda “27976” 
2x16 LCD Serial: (con luz de fondo): búsqueda “27977” 
4x20 LCD Serial: (con luz de fondo) búsqueda “27979” 
 

Características 

• Despliega el juego de caracteres ASCII directamente a la pantalla 
• Ajusta a la siguiente línea automáticamente para fácil despliegue de cadenas de 

texto 
• Trabaja a 2400, 9600, y 19,200 baud 
• Muevee el cursor a cualquier punto en la pantalla con un solo comando 
• Limpia toda la pantalla con un solo comando 
• Le permite definer hasta 8 caracteres personalizados 

Conexiones 

Conectar la LCD Serial al BASIC Stamp es una operation directa, requiere solo 3 pines 
de E/S.  Véa la figura B-1 y B-2 para detalles de conexiones electricas. Vea las Figuras 
B-3 y B-4 en las siguienes páginas para etalles de tamaño y montaje mecánico. 
 

 

Figura B-1 
 
Conexiones de la 
LCD Serial para la 
Rev D y 
anteriores. 

Figura B-2 
 
Conexiones de la 
LCD Serial para la 
Rev E y 
posteriores. 

 
La tabla abajo lista los requerimientos de voltaje y corriente de la LCD Serial, con la luz 
de fondo encendida y apagada. Puesto que en cada caso la corriente que toma excede la 
capacidad de todos los módulos BASIC Stamp, debe energizar la LCD Serial de una 
fuente externa de 5 V. Asegúrese de que la fuente tiene un nivel de corriente adecuado 
para energizar la LCD Serial y el BASIC Stamp. 
 

Estado de la LCD Serial  Voltaje Corriente 
Todos los modelos, luz de fondo apagada 5 VDC 20 mA 

Luz de fondo encendida 27977 / 27979 5 VDC 80 mA 
 

Figura B-3  Especificaciones de Tamaño y Montaje para los modelos 27976, 27977. 
Precaucion: Las dimensiones pueden variar. 

Figura B-4  Especificaciones de Tamaño y Montaje para el modelo 27979. Precaucion: Las 
dimensiones pueden variar. 
 

  

Notas Técnicas  
Característica Descripción 

Tipo de pantalla LCD STN, YG, positive, transflective 
Dirección de observación  6H 
Luz de fondo YG LED 
Temperatura de Operación  -4°F~158°F (-20°C~70°C) 
Temperatura de almacenaje -22°F~176°F (-30°C~80°C) 
Tolerancia dimensional  ±.02” (.5mm) 

Configuración de la velocidad Baud 

Después de conectar la LCD Serial, necesitará seleccionar la velocidad baud a la cual va 
a enviar datos. Tiene 3 opciones:  2400, 9600, y 19,200 baud.  Para establecer esta 
velocidad, mueva los interruptores dip switch en la parte posterior de la LCD a las 
posiciones correctas de acuerdo a la tabla cercana a los interruptores, que aquí se copia: 
 

MODE SW1 SW2 
Test OFF OFF 
2,400 ON OFF 
9,600 OFF ON 
19.200 ON ON  

Como puede ver, hay una cuarta opción llamada Test. Ya que conectó la energía, use este 
modo para confirmar que están correctamente conectados la energía y tierra a la LCD 
antes de que envíe algún dato. Mueva los interruptores a Test y encienda la energía. La 
pantalla LCD se debe encender con la luz de fondo encendida (modelos 27977, 27979) y 
desplegando el siguiente texto:  
 

Parallax, Inc.   
www.parallax.com 

 
Si no ve esto al principio, trate ajustar el contraste de la LCD girando el potenciómetro 
“Increase Contrast” con un desarmador.  Gírelo en dirección de la flecha para hacer que 
los caracteres se muestren más claros. Si aún no los ve, revise sus conexiones eléctricas e 
intente de nuevo. Una vez que haya completado exitosamente el modo prueba, mueva los 
interruptores a las posiciones correctas para seleccionar la velocidad baud a la que quiere 
para su aplicación.  

Displegando Texto 

Ahora que la LCD está configurada, es hora de empesar a enviar texto a la pantalla. Para 
desplegar un caracter de texto en la LCD Serial, simplemente envíe el código ASCII de 
ese caracter a la LCD Serial por el puerto serial a la velcidad baud correcta.   
 
Cuando un caracter es recibido, la LCD Serial despliega ese caracter en la posición actual 
del cursor y luego mueve el cursor una posición a la derecha. Cuando enciende la LCD, 
el cursor esta en la posición extrema izquierda de la línea superior, como podría esperar.  
La pequeña barra en la parte baja del caracter muestra dónde está posicionado el cursor. 
 
Una vez que ha enviado una línea completa de caracteres a la LCD, notará que el cursor 
automáticamente ajusta a la posición extrema izquierda de la segunda línea, como el texto 
en un libro. El texto se ajustará así al final de cada línea, con el final de la línea inferior 
ajustándose a la línea superior de la LCD. El texto nunca “saldrá” de la pantalla; siempre 
verá todos los caracteres que envíe.   
 
Intente el siguiente código en su BASIC Stamp 2 para enviar una cadena de texto a la 
pantalla LCD. Primero, configure la velocidad baud en su LCD Serial a 19,200.  Luego, 
cargue el código siguiente en su BASIC Stamp 2 y córralo. Verá la cadena de texto 
aparecer y ajustarse a la segunda línea de la pantalla.   

En todo su código para la LCD Serial, debe hacer una pausa de 100 ms al inicio para 
darle tiempo para initializar. También debe configurar el pin de Puerto serie en el BASIC 
Stamp a ALTO antes del retraso de 100 ms para arranque, dado que este es el estado 
normal de un puerto serie cuando no está enviando ningún dato. 
 
' {$STAMP BS2} 
 
TxPin           CON     0 
Baud19200       CON     32 
 
  HIGH TxPin                      ' Establece pin alto para ser Puerto serie 
  PAUSE 100                       ' Pause para inicializar la LCD Serial 
  SEROUT TxPin, Baud19200, ["Hello, this text will wrap."] 

 

Moviendo el Cursor 

Cuando envía un caracter a la LCD Serial, siempre se despliega en la posición acual del 
cursor. Hay algunos modos diferentes para mover el cursor en la pantalla  LCD Serial.  
Después de cada caracter que envía, el cursor automáticamente se mueve una posición.  
Junto con esto, hay un juego estándar de comandos para movimiento de que incluye 
Backspace, Carriage Return, y Line Feed. 
 
El comando Backspace/Left (Dec 8) mueve el cursor una posición a la izquierda y el 
comando Right (Dec 9) mueve el cursor una posición a la derecha. Esto puede ser útil 
para mover el cursor para sobreescribir texto existente. Estos comandos ajustan a la 
siguiente line de la pantalla si es necesario.  El comando Line Feed (Dec 10) mueve el 
cursor a la siguiente línea de la pantalla sin cambiar la posición horizontal del cursor.  El 
comando Carriage Return (Dec 13) también mueve el cursor a la siguiente línea, pero 
también mueve el cursor a la posición extrema izquierda en esa línea. El comando Form 
Feed (Dec 12) limpia toda la pantalla y mueve el cursor a la posición extrema izquierda 
en la Línea 0, como cuando enciende la pantalla. Necesitará hacer una pausa de 5mS en 
su código después de enviar el comando Form Feed, para darle a la LCD Serial tiempo 
para limpiar la pantalla. A excepción de Form Feed, ninguno de estos de comandos 
movimiento afecta los caracteres en la pantalla. 
 
También hay  comandos de movimiento directo que puede usar para mover el cursor a 
cualquier posición en la pantalla con un solo comando. Los comandos en el rango Dec 
128 a 143 y Dec 148 a 163 mueven el cursor a las 16 diferentes posiciones en cada una 
de las 2 líneas de los modelos 27976 y 27977. Los comandos en el rango Dec 128 a 207 

mueven el cursor a las 20 diferentes posiciones en cada una de las 4 líneas del modelo 
27979 LCD. 

Controlando la Pantalla 

También tiene control sobre los diversos modos de pantalla de la LCD Serial. El 
comando display-off (Dec 21) apaga la pantalla y todos los caracteres desaparecen. Los 
caracteres no son borrados de la pantalla, e incluso puede escribir nuevos caracteres a la 
pantalla cuando está apagada. Un truco para hacer que mucho texto aparezca de golpe, 
aún a una baja velocidad baud rate, es apagar la pantalla y enviar todo su texto. Entonces, 
cuando la encienda de nuevo, todo el texto aparece instantaneamente. 
 
Los comandos display-on (DEC 22 to 25) encienden nuevamente la pantalla y también 
controlan si desea o no desplegar el cursor y/o hacer que el caracter del cursor parpadee.  
El cursor es la pequeña barra que se muestra abajo del caracter en la at posición actual del 
cursor. La opción de parpadeo hace que ese caracter encienda y apague repetidamente.  
Puede encender o apagar el cursor o su parpadeo en cualquier combinación como se lista 
en la tabla de juego de comandos. Puede cambiar el modo de cursor y parpadeo aún si la 
pantalla ya está encendida; no necesita apagarla y volverla a encender.  
 
Con los modelos 27977 y 27979, puede también controlar la luz de fondo de la pantalla.  
Esta ilumina la pantalla para facilitar su lectura en la oscuridad. Hay comandos para 
encenderla (Dec 17) y apagarla (Dec 18). 

Caracteres personalizados 

La LCD Serial tiene la capacidad de guardar hasta 8 caracteres personalizados definidos 
por el usuario. Los caracteres personalizados son guardados en RAM y por ello necesitan 
ser redefinidos si desconecta la energía. Puede despegar los caracteres personalizados 
enviando los comandos Dec 0 a 7, como se muestra en la tabla de juego de comandos.  El 
caracter personalizado se desplegará en la posición actual del cursor.   
 
Los caracteres personalizados son de 5 pixeles de ancho por 8 pixeles de alto. Cada uno 
de los caracteres es guardado como una serie de datos de 8 bytes donde los 5 bits más 
bajos de cada byte representan una linea de pixeles en el caracter. Los 3 bits más altos de 
cada bytes son ignorados. Un valor 1 en el bit enciende ese pixel (lo hace negro).  La 
línea de pixeles de hasta abajo se deja normalmente en blanco (todo ceros) para poder ver 
el cursor. 

Para definir un caracter personalizado, enviará un total de 9 bytes a la LCD Serial. El 
primer byte necesita ser un comando valido define-caracter-personalizado (Dec 248 a 
255) y debe ser seguido por 8 bytes de datos que definen los pixeles del caracter. La  
LCD serial siempre usará los siguientes 8 bytes que reciba para configurar los pixeles del 
caracter. Los bytes de datos definen el caracter empezando en el renglón de pixeles 
extremo superior, como se muestra en el código ejemplo.   
 
Defina un caracter personalizado usando el código ejemplo siguiente. Primero, configure 
la velocidad baud en su LCD Serial a 19,200. Luego, cargue el siguiente código en su 
BASIC Stamp 2 y córralo. Verá un caracter de diamante aparecer en la pantalla. 
 
' {$STAMP BS2} 
 
TxPin           CON     0 
Baud19200       CON     32 
 
  HIGH TxPin                 ' Establece pin en alto para ser un puerto serial  
  PAUSE 100                       ' Pausa para que LCD Serial inicialize 
 
  SEROUT TxPin, Baud19200, [250]  ' Define caracter personalizado 2 
                                  ' Ahora envia los 8 bytes de datos  
  SEROUT TxPin, Baud19200, [0]    ' 0  = %00000 
  SEROUT TxPin, Baud19200, [4]    ' 4  = %00100      * 
  SEROUT TxPin, Baud19200, [14]   ' 14 = %01110    * * * 
  SEROUT TxPin, Baud19200, [31]   ' 31 = %11111  * * * * * 
  SEROUT TxPin, Baud19200, [14]   ' 14 = %01110    * * * 
  SEROUT TxPin, Baud19200, [4]    ' 4  = %00100      * 
  SEROUT TxPin, Baud19200, [0]    ' 0  = %00000 
  SEROUT TxPin, Baud19200, [0]    ' 0  = %00000 
  SEROUT TxPin, Baud19200, [2]   ' Despliega el nuevo caracter personalizado 2 

 

Conjunto de Comandos 

Las tablas a continuación listan todos los comandos válidos de la LCD Serial. Los 
comandos marcados como N/A son invalidos y son ignorados. Las lineas de la pantalla 
LCD están numeradas desde 0, siendo la Linea 0 la de hasta arriba. Las posiciones de 
caracter en cada línea están numeradas empezando desde 0, siendo la posición 0 la de la 
extrema izquierda en la línea. 

Dec Hex Acción 
0 00 Despliega el caracter personalizado 0 
1 01 Despliega el caracter personalizado 1 
2 02 Despliega el caracter personalizado 2 
3 03 Despliega el caracter personalizado 3 
4 04 Despliega el caracter personalizado 4 
5 05 Despliega el caracter personalizado 5 
6 06 Despliega el caracter personalizado 6 
7 07 Despliega el caracter personalizado 7 

8 08 Backspace / Left - El cursor es movido una posición a la izquierda. El comando no 
borra el caracter. 

9 09 Right - El cursor es movido una posición a la derecha. El comando no borra el 
caracter. 

10 0A 
Line Feed - El cursor es movido hacia abajo una línea. Para los modelos LCD de 2 
líneas, si está en la línea 0 va a la línea1.  Si está en la línea 1, se ajusta a la línea 
0. La posición horizontal se mantiene igual. 

11 0B N/A 

12 0C Form Feed - El cursor es movido a la posición 0 en la línea 0 y toda la pantalla es 
limpiada. Los usuarios deben hacer una pausa de 5mS depués de este comando. 

13 0D Carriage Return – En LCDs de 2 líneas, si está en línea 0 el cursor se mueve a la 
posición 0 en la linea 1. Si está en la línea 1 se ajusta a la posición 0 en la línea 0. 

14 - 16 0E - 10 N/A 
17 11 Enciende la pantalla (solo los modelos 27977, 27979) 
18 12 Apaga la pantalla (Default) 
19 - 20 13 - 14 N/A 
21 15 Apaga la pantalla 
22 16 Enciende la pantalla, con cursor apagado y sin parpadeo 
23 17 Enciende la pantalla, con cursor apagado y con parpaedo de caracter 
24 18 Enciende la pantalla, con cursor encendido y sin parpadeo (Default) 
25 19 Turn the display on, con cursor encendido y con parpaedo de caracter 
26 - 31 1A - 1F N/A 
32 - 127 20 - 7F Despliega caracteres ASCII.  Vea la tabla de juego de caracteres ASCII. 
128 80 Mueve el cursor a la línea 0, posición 0 
129 81 Mueve el cursor a la línea 0, posición 1 
130 82 Mueve el cursor a la línea 0, posición 2 
131 83 Mueve el cursor a la línea 0, posición 3 
132 84 Mueve el cursor a la línea 0, posición 4 
133 85 Mueve el cursor a la línea 0, posición 5 
134 86 Mueve el cursor a la línea 0, posición 6 

Dec Hex Action 
135 87 Mueve el cursor a la línea 0, posición 7 
136 88 Mueve el cursor a la línea 0, posición 8 
137 89 Mueve el cursor a la línea 0, posición 9 
138 8A Mueve el cursor a la línea 0, posición 10 
139 8B Mueve el cursor a la línea 0, posición 11 
140 8C Mueve el cursor a la línea 0, posición 12 
141 8D Mueve el cursor a la línea 0, posición 13 
142 8E Mueve el cursor a la línea 0, posición 14 
143 8F Mueve el cursor a la línea 0, posición 15 
144 90 Mueve el cursor a la línea 0, posición 16 (solo el modelo 27979) 
145 91 Mueve el cursor a la línea 0, posición 17 (solo el modelo 27979) 
146 92 Mueve el cursor a la línea 0, posición 18 (solo el modelo 27979) 
147 93 Mueve el cursor a la línea 0, posición 19 (solo el modelo 27979) 
148 94 Mueve el cursor a la línea 1, posición 0 
149 95 Mueve el cursor a la línea 1, posición 1 
150 96 Mueve el cursor a la línea 1, posición 2 
151 97 Mueve el cursor a la línea 1, posición 3 
152 98 Mueve el cursor a la línea 1, posición 4 
153 99 Mueve el cursor a la línea 1, posición 5 
154 9A Mueve el cursor a la línea 1, posición 6 
155 9B Mueve el cursor a la línea 1, posición 7 
156 9C Mueve el cursor a la línea 1, posición 8 
157 9D Mueve el cursor a la línea 1, posición 9 
158 9E Mueve el cursor a la línea 1, posición 10 
159 9F Mueve el cursor a la línea 1, posición 11 
160 A0 Mueve el cursor a la línea 1, posición 12 
161 A1 Mueve el cursor a la línea 1, posición 13 
162 A2 Mueve el cursor a la línea 1, posición 14 
163 A3 Mueve el cursor a la línea 1, posición 15 
164 A4 Mueve el cursor a la línea 1, posición 16 (solo el modelo 27979) 
165 A5 Mueve el cursor a la línea 1, posición 17 (solo el modelo 27979) 
166 A6 Mueve el cursor a la línea 1, posición 18 (solo el modelo 27979) 
167 A7 Mueve el cursor a la línea 1, posición 19 (solo el modelo 27979) 
168 A8 Mueve el cursor a la línea 2, posición 0 (solo el modelo 27979) 
169 A9 Mueve el cursor a la línea 2, posición 1 (solo el modelo 27979) 
170 AA Mueve el cursor a la línea 2, posición 2 (solo el modelo 27979) 
171 AB Mueve el cursor a la línea 2, posición 3 (solo el modelo 27979) 
172 AC Mueve el cursor a la línea 2, posición 4 (solo el modelo 27979) 
173 AD Mueve el cursor a la línea 2, posición 5 (solo el modelo 27979) 
174 AE Mueve el cursor a la línea 2, posición 6 (solo el modelo 27979) 
175 AF Mueve el cursor a la línea 2, posición 7 (solo el modelo 27979) 
176 B0 Mueve el cursor a la línea 2, posición 8 (solo el modelo 27979) 
177 B1 Mueve el cursor a la línea 2, posición 9 (solo el modelo 27979) 
178 B2 Mueve el cursor a la línea 2, posición 10 (solo el modelo 27979) 
179 B3 Mueve el cursor a la línea 2, posición 11 (solo el modelo 27979) 
180 B4 Mueve el cursor a la línea 2, posición 12 (solo el modelo 27979) 

Dec Hex Action 
181 B5 Mueve el cursor a la línea 2, posición 13 (solo el modelo 27979) 
182 B6 Mueve el cursor a la línea 2, posición 14 (solo el modelo 27979) 
183 B7 Mueve el cursor a la línea 2, posición 15 (solo el modelo 27979) 
184 B8 Mueve el cursor a la línea 2, posición 16 (solo el modelo 27979) 
185 B9 Mueve el cursor a la línea 2, posición 17 (solo el modelo 27979) 
186 BA Mueve el cursor a la línea 2, posición 18 (solo el modelo 27979) 
187 BB Mueve el cursor a la línea 2, posición 19 (solo el modelo 27979) 
188 BC Mueve el cursor a la línea 3, posición 0 (solo el modelo 27979) 
189 BD Mueve el cursor a la línea 3, posición 1 (solo el modelo 27979) 
190 BE Mueve el cursor a la línea 3, posición 2 (solo el modelo 27979) 
191 BF Mueve el cursor a la línea 3, posición 3 (solo el modelo 27979) 
192 C0 Mueve el cursor a la línea 3, posición 4 (solo el modelo 27979) 
193 C1 Mueve el cursor a la línea 3, posición 5 (solo el modelo 27979) 
194 C2 Mueve el cursor a la línea 3, posición 6 (solo el modelo 27979) 
195 C3 Mueve el cursor a la línea 3, posición 7 (solo el modelo 27979) 
196 C4 Mueve el cursor a la línea 3, posición 8 (solo el modelo 27979) 
197 C5 Mueve el cursor a la línea 3, posición 9 (solo el modelo 27979) 
198 C6 Mueve el cursor a la línea 3, posición 10 (solo el modelo 27979) 
199 C7 Mueve el cursor a la línea 3, posición 11 (solo el modelo 27979) 
200 C8 Mueve el cursor a la línea 3, posición 12 (solo el modelo 27979) 
201 C9 Mueve el cursor a la línea 3, posición 13 (solo el modelo 27979) 
202 CA Mueve el cursor a la línea 3, posición 14 (solo el modelo 27979) 
203 CB Mueve el cursor a la línea 3, posición 15 (solo el modelo 27979) 
204 CC Mueve el cursor a la línea 3, posición 16 (solo el modelo 27979) 
205 CD Mueve el cursor a la línea 3, posición 17 (solo el modelo 27979) 
206 CE Mueve el cursor a la línea 3, posición 18 (solo el modelo 27979) 
207 CF Mueve el cursor a la línea 3, posición 19 (solo el modelo 27979) 
208 - 247 D0 – F7 N/A 
248 F8 Define caracter personalizado 0.  8 bytes de datos deben seguir al comando.  
249 F9 Define caracter personalizado 1.  8 bytes de datos deben seguir al comando. 
250 FA Define caracter personalizado 2.  8 bytes de datos deben seguir al comando. 
251 FB Define caracter personalizado 3.  8 bytes de datos deben seguir al comando. 
252 FC Define caracter personalizado 4.  8 bytes de datos deben seguir al comando. 
253 FD Define caracter personalizado 5.  8 bytes de datos deben seguir al comando. 
254 FE Define caracter personalizado 6.  8 bytes de datos deben seguir al comando. 
255 FF Define caracter personalizado 7.  8 bytes de datos deben seguir al comando. 

Juego de Caracteres ASCII  

La siguiente tabla muestra todos los caracteres ASCII como se despliegan en la LCD 
Serial. Todos los caracteres ASCII (Dec 32 to 127) son caracteres ASCII estandar, 
excepto por la contradiagonal ‘\’ (Dec 92) y el tilde ‘~’ (Dec 126).  Para su facilidad, la 
LCD Serial viene pre-programada con estos caracteres en los primeros 2 caracteres 
personalizados. Entonces, para desplegar una contradiagonal, use el comando Dec 0 y 
para desplegar un tilde, use el comando Dec 1. Por supuesto, siempre puede sobreescribir 
estos caracteres con sus propios caracteres personalizados. 

Apéndice C: Definiciones de Caracteres 
Hexadecimales  
 
Muchas  notas de aplicaciones de la LCD y documentación usan números hexadecimales 
en vez de números binarios para definir comandos y caracteres. En PBASIC, escribir un 
signo de exclamación es tan simple como un comando SEROUT con un signo de 
exclamaciío entre comillas. 

 
SEROUT 14, 84, ["!"] 
 

No todos los lenguajes de programación para controladores soportan este uso nativo de 
caracteres escribibles. En algunos casos, sobre todo en lenguaje ensamblador, se usa el 
código ASCII para el signo de exclamación. El código ASCII para este signo es 33, y aún 
en PBASIC, el comando SEROUT 14, 84, [33] realiza la misma tarea. En lenguaje 
ensamblador, los valores hexadecimales son algunas veces el número base preferido 
porque hace ciertas tareas más fáciles. Debido a esto, mucha documentación para LCD 
lista sus comandos LCD como valores hexadecimales.   
 
The equivalente hexadecimal de decimal-33 es hexadecimal-21. Esto es (2 × 16) + 1.  
Puede usar el operator $ para especificar que un valor es hexadecimal, lo que haría que el 
comando para desplegar un signo de exclamación fuera: SEROUT 14, 84, [$21]. 
 
He aquí un ejemplo de un comando SEROUT que define una barra en una gráfica de barra. 
Llena la parte de abajo del caracter con pixeles negros y deja la parte mitad superior en 
blanco: 
 
SEROUT 14, 84, [250, 
                %00000,           ' 
                %00000,           ' 
                %00000,           ' 
                %00000,           ' 
                %11111,           ' * * * * * 
                %11111,           ' * * * * * 
                %11111,           ' * * * * * 
                %11111]           ' * * * * * 
 

He aquí un comando equivalente usando valores hexadecimales. Si bien salva mucho 
espacio, no es tan fácil de entender como definir un caracter personalizado con números 
binarios. Siendo así, verá programas de aplicaciones PBASIC escritos de esta manera de 

vez en vez, principalente por la prevalecencia de valores hexadecimal en documentación 
para LCD.  
 
SEROUT 14, 84, [250, $00, $00, $00, $00, $1F, $1F, $1F, $1F] 
 

La Table 7-1 cuenta a 15 en decimal, hexadecimal y binario. En términos de convertir de 
hexadecimal a binary y viceversa, esta tabla es todo lo que necesitará. La razón es porque 
cada dígito hexadecimal suelto corresponde a un grupo de 4 dígitos binarios.   
 

Table 7-1: Valores Decimales, Hexadecimales y Binarios  
Base 

10 0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 

Base 
16 0 1 2 3 4 5 6 7 8 9 A B C D E F 

Base 
2 

00
00 

00
01 

00
10 

00
11 

01
00 

01
01 

01
10 

01
11 

10
00 

10
01 

10
10 

10
11 

11
00 

11
01 

11
10 

11
11 

 
Ejemplo: Convierta $8FE6 a binario. 
Solución: Cada dígito hexadecimal se convierte en un grupo de 4 números binarios, 
entonces el valor binario puede ser escrito con la ayuda de la Table 7-1: 
 
     Hexadecimal         8          F         E          6 
     Binario               1000    1111    1110    0110 
 
     $8FE6 = %1000111111100110 
 
Ejemplo: Convierta %1100011000001001 a hexadecimal. 
Solución: Arregle el número binario en grupos de 4 dígitos, luego use la Table 7-1: 
 
     Binario               1100    0110    0000    1001 
     Hexadecimal         C         6          0          9 
 
     %1100011000001001 = $C609 

Programa Ejemplo: ConvertBinaryToHexadecimal.bs2 

Por supuesto, puede también hacer que el BASIC Stamp lo haga por usted. Simplemente 
use el operador % para definir el número binario y entonces usar el formateador HEX en 
un comando SEROUT para desplegar el valor. 

√ Inténtelo y vea si la LCD coincide con nuestros cálculos   
 
' Sensores Inteligentes y sus Aplicaciones - ConvertBinaryToHexadecimal.bs2 
 
' {$STAMP BS2} 
' {$PBASIC 2.5} 
 
PAUSE 250                                    ' Estabiliza la fuente de energia 
SEROUT 14, 84, [22, 12]                      ' Enciende y limpia la pantalla 
PAUSE 5                                      ' Retraso de 5 ms para limpiar 
 
SEROUT 14, 84, ["Value = ", HEX %1100011000001001] 
 
END 
 

Su Turno – Convierta de Hexadecimal a Binario 

Convertir de hexadecimal a binario es un asunto de usar el formateador BIN en vez de 
HEX y using el operator $ para decirle al Editor del BASIC Stamp que le está dando un 
valor hexadecimal (en vez de % para binario). Si está conviertiendo 4 dígitos 
hexadecimal a binario, el resultado será un número de 16 dígitos. Entonces, el comando 
SEROUT también debe ser modificado para desplegar el resultado comenzando al incio de 
la segunda línea. 
 

√ Comente la línea de código existente que ejecuta la conversion de binario a 
hexadecimal. 
 
SEROUT 14, 84, ["Value = ", 148, BIN $8FE6] 

Apéndice D: Listado de Partes  
 
Requerimientos del sistema de Cómputo : 

• PC con Windows 2000/XP  
• Un puerto serie disponible o puerto USB. Si necesita un adaptador USB a 

Serie, recomendamos el Parallax parte #800-00030. 
• Acceso a Internet  

 
Requerimientos de Software:  

• Editor BASIC Stamp para Windows v2.0 o más reciente (Descarga gratis en 
www.parallax.com) 

• Programas ejemplo seleccionados (Descarga gratis en www.parallax.com) 
•  Microsoft Notepad y Microsoft Excel 2002 o más reciente (para estudios de 

aceleración del Capítulo 6)  
 
Requerimientos de Hardware: 

• Uno de los siguientes kits que incluye una plataforma de programación BASIC 
Stamp 2, más la fuente de energía adecuada o las baterías: 
o Board of Education Full Kit Serial (#28102) o USB (#28802)  
o Boe-Bot Robot Kit  Serial (#28132) o USB (#28832) 
o BASIC Stamp activity Kit* (incluye HomeWork Board) (#90005) 

• Sensores Inteligentes y sus Aplicaciones, Partes y Texto (#28029) o Kit de 
partes (#130-28029) 

 
Items de casa requeridos: 

• Pequeña barra magnética 
• Brújula Mecánica (para calibrar el Modulo de brújula, Capítulo 4) 
• Carro RC y  Controlador, con baterías (para Capítulo 6, actividad #5) 
• Rueda de bicicleta (para Capítulo 6, actividad #6) 

 
*Todas las actividades en este texto son compatibles con la tarjeta BASIC Stamp 
HomeWork mientras que esté usando la LCD sin luz de fondo incluída en el kit de partes 
de Sensores Inteligentes. Si está usando una LCD con luz de fondo, use también una 
tarjeta Board of Education para proteger su LCD.   

Sensores Inteligentes y sus Aplicaciones, Partes y Texto  #28029 (Sin libro, #130-28029) 
Partes y  cantidades sujetas a cambio sin previo aviso 

Parallax Parte # Descripción Cantidad 
122-28029 Libro Sensores Inteligentes y sus Aplicaciones 1 
150-01020 Resistenciass 220 Ω, ¼ watt 5% 2 

27976 LCD Serial Parallax 2 x 16 sin luz de fondo 1 
28015 Sensor de Distancia Ultrasónico Ping)))  1 
28017 Acelerómetro Memsic Bi-axial  1 
29123 Módulo de brújula Hitachi  1 

451-00303 Conector de 3 pines macho/macho 3 
700-00003 4-40 tuercas zincadas-plateadas 6 
700-00028 Tornillos de 1/4” 4-40 pan-head 4 
710-00006 Tornillos de1/2“ 4-40 pan-head 2 
713-00005 Espaciador #4 de 1/4” circular de nylon 2 
720-00011 Herrajes para montaje universal de 90 grados 4 
800-00016 Bolsa con 10 cables conectores de 3 pulgadas  2 
805-00002 Cable Extension para Servo/LCD  2 
900-00001 Piezo parlante 1 

 

Puede encontrar algunos componentes electrónicos extra en su kit que no se usan en el libro 
Sensores Inteligentes y sus Aplicaciones. Si quiere usar estas para construir sus propios 
circuitos, aprenda acerca los capacitores electrolíticos y sus requerimientos de seguridad 
antes de proceder PRECAUCIÓN: El uso incorrecto de capacitores electrolíticos puede 
causar que exploten. Siga las siguienes reglas de seguridad para evitar cualquier daño 

Los capacitores electrolíticos tienen una terminal positiva (+) y una 
negativa (-). El voltaje en la terminal (+) del capacitor debe ser siempre mayor 
que el voltaje en su terminal (-). Use la imagen (derecha) para identificar las 
terminales (+) y (-). Siempre asegúrese de conectar estas terminales 
exactamente como se muestra en diagramas confiables. Conectar uno de 
estos capacitores incorrectamente puede dañarlo. En algunos circuitos 
conectar este tipo de capacitor incorrectamente y luego energizar puede 
causar su ruptura e incluso explotar. Vss es el voltaje más bajo (0 V) en las 
tarjetas Board of Education y BASIC Stamp HomeWork. Al conectar la 
terminal negativa del capacitor a Vss asegurará que la polaridad a través de 
las terminales del capacitor siempre será correcta. 

 

SEGURIDAD  
Siempre apague la energía antes de construir o modificar los circuitos. 
Siempre observe la polaridad cuando conecte capacitores electrolíticos.  
Nunca invierta la polaridad de la fuente en cualquier capacitor polar.  
Use lentes de seguridad cuando use estos capacitores.         
Mantenga sus manos y cara lejos de estos capacitores al energizarlos. 

Símbolo del 
Capacitor  

+-

10 µF

